

## MATH 8020 California House of Representatives and Committee Membership

The Excel file California House of Representatives contains a list of the 2018 elected representatives and committees on which they serve (along with some additional unneeded information).

Obs	District	Name	Party	Office_Room	Phone	Committee_Assignment
1	1st	LaMalfa, Doug	R	322 CHOB	(202) 225-3076	Agriculture
2						Natural Resources
3						Transportation and Infrastructure
4						
5	2nd	Huffman, Jared	D	1406 LHOB	(202) 225-5161	Natural Resources
6						Transportation and Infrastructure
7						
8	3rd	Garamendi, John	D	2438 RHOB	(202) 225-1880	Armed Services
9						Transportation and Infrastructure
10						

After some data wrangling, we get the following.

Obs	District	Name	Committee_Assignment
1	1st	LaMalfa, Doug	Agriculture
2		LaMalfa, Doug	Natural Resources
3		LaMalfa, Doug	Transportation and Infrastructure
4	2nd	Huffman, Jared	Natural Resources
5		Huffman, Jared	Transportation and Infrastructure
6	3rd	Garamendi, John	Armed Services
7		Garamendi, John	Transportation and Infrastructure
8	4th	McClintock, Tom	Budget
9		McClintock, Tom	Natural Resources
10	5th	Thompson, Mike	Ways and Means

The first proc optgraph call provides a summary of the graph. The graph is connected with density 0.038. Vertices of degree 1 indicate a member who sits on a single committee or committee with only one representative from CA serving.

The second proc optgraph call determines the set of cutvertices in the graph.

The third proc optgraph call compute the degree of each vertex. Since this is n unweighted graph, influence in the graph is determined strictly by vertex degree. The Natural Resources committee has the most CA reps at 11. Zoe Lofgren serves on the most committees at 4.

```
/* Generated Code (IMPORT) */
/* Source File: California House of Representatives.xlsx */
/* Source Path: /gpfs/user_home/jdemaio/MATH 8020 */
/* Code generated on: 1/30/18, 1:25 PM */

%web_drop_table(WORK.IMPORT);

FILENAME REFFILE '/gpfs/user_home/jdemaio/MATH 8020/California House
of Representatives.xlsx';

PROC IMPORT DATAFILE=REFFILE
      DBMS=XLSX
      OUT=WORK.IMPORT;
      GETNAMES=YES;
RUN;

PROC print DATA=WORK.IMPORT (obs=10);
RUN;

DATA CA_Graph (drop=party phone Office_Room); SET WORK.IMPORT;
if cmiss(Committee_Assignment) then delete;
retain temp_name;
      if not missing(name) then temp_name=name;
      else name=temp_name;
      drop temp_name;
run;

PROC print DATA=CA_Graph (obs=10);
RUN;

proc optgraph
data_links = CA_Graph
out_nodes = CA_Graph_Nodes
out_links = CA_Graph_links;
data_links_var
from = name
to = Committee_Assignment;
summary
concomp
out = CA_Graph_Summary1;
run;

proc optgraph
data_links = CA_Graph
```

```
out_nodes= CA_cutvertices;
data_links_var
from = name
to = Committee_Assignment
;
biconcomp; /*find cutvertices in graph*/
run;

proc optgraph
data_links = CA_Graph
out_nodes= CA_vertex_degrees; /* computes degree of each vertex */
data_links_var
from = name
to = Committee_Assignment
;
centrality degree = out;
run;

%web_open_table(WORK.IMPORT);
```